

On $\int \mathrm{td}_X^{1/2}$ of O’Grady’s exceptional hyperkähler varieties

Pieter Belmans

July 6, 2021

Abstract

In [2, Example 5.6(4)] (the first version) an exercise was left to the reader, to determine $\int \mathrm{td}_X^{1/2}$ for O’Grady’s 6- resp. 10-dimensional hyperkähler varieties. The values turn out to be $\frac{2}{3}$ resp. $\frac{4}{15}$.

These coincide with the numbers for Kum₃- resp. K3^[5]-type. After informing Chen Jiang of this he gave an abstract proof using the equality of Riemann–Roch polynomials, see [3, Lemma 5.7] (the second version).

In this note I detail the explicit computation, along with the code to do it, in the hope that it (in particular the code) is useful to someone.

1 On $\int \mathrm{td}_X^{1/2}$ of hyperkähler varieties

The following is a combination of the lower bound from [1] and the upper bound from [2, Corollary 5.5].

Theorem 1 (Hitchin–Sawon, Jiang). Let X be a hyperkähler variety of dimension $2n > 2$. Then

$$(1) \quad 0 < \int \mathrm{td}_X^{1/2} < 1.$$

For $n = 1$ we have that $\int \mathrm{td}_{K3} = 1$. The values for the infinite families of hyperkähler varieties are known, and given in [8, Proposition 19 and 21].

Proposition 2 (Sawon). We have

$$(2) \quad \int \mathrm{td}_{K3^{[n]}} = \frac{(n+3)^n}{4^n n!},$$

$$(3) \quad \int \mathrm{td}_{\mathrm{Kum}^n} = \frac{(n+1)^{n+1}}{4^n n!}.$$

The new¹ result in this note is the computation of $\int \mathrm{td}_X^{1/2}$ for O’Grady’s 6- and 10-dimensional hyperkähler varieties; per the exercise left to the reader in [2, Example 5.6(4)] (the first version).

commit: 399ac71 date: 2021-07-06 11:19:02 +0200 (HEAD -> master)

¹After sharing the first version of this note with the author of [2] he was able to give a proof based on the equality of Riemann–Roch polynomials from [7], see [3, Lemma 5.7] (the second version).

	c_2^3	$c_2 c_4$	c_6
$K3^{[3]}$	36800	14720	3200
Kum^3	30208	6784	448
OG6	30720	7680	1920

Table 1: Chern numbers for 6-dimensional hyperkähler varieties

Proposition 3. We have

$$(4) \quad \int \mathrm{td}_{\mathrm{OG6}}^{1/2} = \frac{2}{3}$$

$$(5) \quad \int \mathrm{td}_{\mathrm{OG10}}^{1/2} = \frac{4}{15}$$

We'll collect some details in the computation section; but it is just an exercise in the explicit expression of the Todd class and expressing the Chern character using Chern classes, performed entirely within Sage.

Evaluating the formulas from Proposition 2 for $n = 3$ and $n = 5$ respectively gives the following (curious?)² coincidence.

Corollary 4. We have the following equalities:

$$(6) \quad \int \mathrm{td}_{\mathrm{OG6}}^{1/2} = \int \mathrm{td}_{\mathrm{Kum}^3}^{1/2}$$

$$(7) \quad \int \mathrm{td}_{\mathrm{OG10}}^{1/2} = \int \mathrm{td}_{K3^{[5]}}^{1/2}$$

The important fact here is that the Chern numbers of these varieties are distinct, see Tables 1 and 2. These are collected from

- [6, page 128] for $K3^{[3]}$ and Kum^3
- [5, Appendix A] for Kum^5
- *I could not find the values for $K3^{[5]}$*
- [4, Proposition 6.8] for OG6
- [7, Appendix A] for OG10

and these moreover allow for an independent check that the intermediate steps of the implementation are correct.

Acknowledgements I thank Chen Jiang for interesting discussions about this integral and finding an abstract proof for the equality, and Olivier Debarre for spotting a typo.

²They are implied by the equality of Riemann–Roch polynomials, so I leave it up to the reader to decide how curious the equality is. The equality of the Riemann–Roch polynomials is proven in a somewhat ad hoc way after all.

	c_2^5	$c_2^3 c_4$	$c_2^2 c_6$	$c_2 c_8$	$c_2 c_4^2$	$c_4 c_6$	c_{10}
K3 ^[5]	?	?	?	?	?	?	?
Kum ⁵	84478464	26220672	8141472	3141504	979776	142560	2592
OG10	127370880	53071200	12383280	1791720	22113000	5159700	176904

Table 2: Chern numbers for 10-dimensional hyperkähler varieties

2 The computation

By [2, §2.4] the square root of the Todd class of a hyperkähler variety X is given by

$$(8) \quad \mathrm{td}_X^{1/2} = \exp \left(- \sum_{k=1}^{\infty} b_{2k}(2k)! \, \mathrm{ch}_{2k}(X) \right),$$

where b_{2k} is the $2k$ th modified Bernoulli number, in the generating function

$$(9) \quad \sum_{k=0}^{+\infty} b_{2k} x^{2k} = \frac{1}{2} \ln \left(\frac{e^{x/2} - e^{-x/2}}{x/2} \right)$$

or in a closed form expression given as

$$(10) \quad b_{2k} := \begin{cases} \frac{\ln(2)}{2} & k = 0 \\ \frac{B_{2k}}{2(2k)^2 \Gamma(2k)} & k \geq 1 \end{cases}$$

where B_{2k} is the $2k$ th Bernoulli number.

The homogeneous part of degree $2n$ in $\mathrm{td}_X^{1/2}$ for a $2n$ -dimensional hyperkähler variety X^{2n} for $n = 3$ (resp. $n = 5$) is given by

$$(11) \quad \begin{aligned} (\mathrm{td}_{X^6}^{1/2})_6 &= -\frac{1}{82944} \mathrm{ch}_2^3 - \frac{1}{5760} \mathrm{ch}_2 \mathrm{ch}_4 - \frac{1}{504} \mathrm{ch}_6 \\ (\mathrm{td}_{X^{10}}^{1/2})_{10} &= -\frac{1}{955514880} \mathrm{ch}_2^5 - \frac{1}{19906560} \mathrm{ch}_2^3 \mathrm{ch}_4 - \frac{1}{2764800} \mathrm{ch}_2 \mathrm{ch}_4^2 - \frac{1}{580608} \mathrm{ch}_2^2 \mathrm{ch}_6 \\ &\quad - \frac{1}{120960} \mathrm{ch}_4 \mathrm{ch}_6 - \frac{1}{11520} \mathrm{ch}_2 \mathrm{ch}_8 - \frac{1}{264} \mathrm{ch}_{10} \end{aligned}$$

so that after the substitution of the products of the ch_{2k} by the Chern numbers we end up with

$$(12) \quad \begin{aligned} (\mathrm{td}_{X^6}^{1/2})_6 &= \frac{31}{967680} c_2^3 - \frac{11}{241920} c_2 c_4 + \frac{1}{60480} c_6 \\ (\mathrm{td}_{X^{10}}^{1/2})_{10} &= \frac{73}{3503554560} c_2^5 - \frac{1073}{15328051200} c_2^3 c_4 + \frac{311}{7664025600} c_2 c_4^2 + \frac{61}{1277337600} c_2^2 c_6 \\ &\quad - \frac{1}{45619200} c_4 c_6 - \frac{53}{1916006400} c_2 c_8 + \frac{1}{95800320} c_{10} \end{aligned}$$

Now we can substitute the Chern numbers and we are done. These computations are all performed in Sage, and one can easily modify the code to see intermediate steps (such as the expression of the Chern character in terms of the Chern classes, etc.)

References

- [1] Nigel Hitchin and Justin Sawon. “Curvature and characteristic numbers of hyper-Kähler manifolds”. In: *Duke Math. J.* 106.3 (2001), pp. 599–615. MR: 1813238.
- [2] Chen Jiang. *Positivity of Riemann–Roch polynomials and Todd classes of hyperkähler manifolds*. arXiv: 2008.04685v1 [math.AG].
- [3] Chen Jiang. *Positivity of Riemann–Roch polynomials and Todd classes of hyperkähler manifolds*. arXiv: 2008.04685v2 [math.AG].
- [4] Giovanni Mongardi, Antonio Rapagnetta, and Giulia Saccà. “The Hodge diamond of O’Grady’s six-dimensional example”. In: *Compos. Math.* 154.5 (2018), pp. 984–1013. MR: 3798592.
- [5] Marc Nieper-Wisskirchen. “On the Chern numbers of generalised Kummer varieties”. In: *Math. Res. Lett.* 9.5-6 (2002), pp. 597–606. MR: 1906063.
- [6] Marc Nieper-Wißkirchen. *Chern numbers and Rozansky-Witten invariants of compact hyper-Kähler manifolds*. World Scientific Publishing Co., Inc., River Edge, NJ, 2004, pp. xxii+150. ISBN: 981-238-851-6. MR: 2110899.
- [7] Ángel David Ríos Ortiz. *Riemann-Roch Polynomials of the known Hyperkähler Manifolds*. arXiv: 2006.09307v2 [math.AG].
- [8] Justin Sawon. *Rozansky-Witten invariants of hyperkähler manifolds*. arXiv: math/0404360v1 [math.DG].

```

1  # expressions taken from Propositions 19 and 21 of Sawon's PhD thesis
2  def K3n(n): return (n+3)^n / (4^n * factorial(n))
3  def Kum_n(n): return (n+1)^(n+1) / (4^n * factorial(n))
4
5  def todd_root(n):
6      """Determine the square root of the Todd class of a hyperkaehler
7
8      Returns the square root of the Todd class expressed in terms of Chern numbers
9      together with the Chern numbers as monomials in a polynomial ring.
10     """
11     def modified_bernoulli(n):
12         if n == 0: return ln(2) / 2
13         if n % 2 == 1: return 0
14         return bernoulli(n) / (2 * n**2 * gamma(n))
15
16     def exp(t, precision=n):
17         return sum([t**i / factorial(i) for i in range(precision + 1)])
18
19     R = PolynomialRing(QQ, ["ch" + str(2*i) for i in range(1, n+1)])
20
21     # (1) truncated power series expansion of square root of Todd class
22     result = prod([exp(-modified_bernoulli(2*k) * factorial(2*k) \
23         * R.gen(k-1)) for k in range(1, n+1)])
24
25     # (2) now collect degree 2*n part
26     # not really Chern numbers but didn't have better name
27     chern_numbers = WeightedIntegerVectors(n, range(1, n+1))
28     chern_numbers = [prod([R.gen(i)**c for (i, c) in enumerate(C)]) for C in chern_numbers]
29     result = sum([result.monomial_coefficient(ch) * ch for ch in chern_numbers])
30
31     # (3) from ch_i to c_i
32     S = PolynomialRing(QQ, ["c" + str(2*i) for i in range(1, n+1)])
33     morphism = []
34
35     for k in range(2, 2*n+1, 2):
36         # go from the Chern character to Chern classes using symmetric functions
37         Sym = SymmetricFunctions(QQ)
38         terms = Sym.elementary()(Sym.powersum()[k]).monomial_coefficients().items()
39
40         # select only the even Chern classes on a hyperkaehler
41         terms = [(E, c) for (E, c) in terms if all([i % 2 == 0 for i in E])]
42
43         expression = 1 / factorial(k) \
44             * sum([c * prod([S.gen(i // 2 - 1) for i in E]) for (E, c) in terms])
45         morphism.append(expression)
46
47     result = R.hom(morphism, S)(result)
48
49     # (4) list the Chern numbers that can appear
50     chern_numbers = WeightedIntegerVectors(n, range(1, n+1))
51     chern_numbers = [prod([S.gen(i)**c for (i, c) in enumerate(C)]) for C in chern_numbers]
52
53     return (result, chern_numbers)
54

```

```

55 print("0'Grady 6-fold")
56 (square_root, chern_numbers) = todd_root(3)
57 square_root.parent().inject_variables()
58 values = {c2^3: 30720, c2*c4: 7680, c6: 1920}
59
60 value = sum([square_root.monomial_coefficient(c) * values[c] for c in values])
61 assert value == Kum_n(3)
62 print(value, "\n")
63
64
65 print("0'Grady 10-fold")
66 (square_root, chern_numbers) = todd_root(5)
67 square_root.parent().inject_variables()
68 values = {c2^5: 127370880, c2^3*c4: 53071200, c2^2*c6: 12383280, c2*c8: 1791720, \
69           c2*c4^2: 22113000, c4*c6: 5159700, c10: 176904}
70
71 value = sum([square_root.monomial_coefficient(c) * values[c] for c in values])
72 assert value == K3n(5)
73 print(value, "\n")
74
75
76 print("Generalised Kummer variety of dimension 6-fold: computation by hand")
77 (square_root, chern_numbers) = todd_root(3)
78 square_root.parent().inject_variables()
79 values = {c2^3: 30208, c2*c4: 6784, c6: 448}
80
81 value = sum([square_root.monomial_coefficient(c) * values[c] for c in values])
82 assert value == Kum_n(3)
83 print(value, "\n")

```